

COME CREARE NUOVI BANTER PER BALDUR'S GATE 2

Paragrafi

[I. Pre-Introduzione](#)

[II. Introduzione](#)

[III.1 I 'Banter'](#)

[III.2 La variabile](#)

[III.3 Le linee di testo](#)

[III.4 L'esempio](#)

[III.5 Il "terzo incomodo"](#)

[III.6 Altri casi di battute particolari](#)

[IV. Gli script](#)

[V. Creare l'installazione](#)

[VI. La traduzione](#)

[VII. Conclusione](#)

[I. Pre-Introduzione](#)

Tutti i termini in inglese o quelli tecnici di particolare importanza sono tra virgolette (es: 'xxx').

[II. Introduzione](#)

Questo tutorial si pone l'obbiettivo di insegnarti a creare velocemente nuovi dialoghi tra i personaggi unibili al gruppo di Baldur's Gate 2 (in gergo tecnico, nuovi 'banter' per gli 'npc') anche senza essersi mai avventurati prima nella modifica del videogioco (tecnicamente, 'modding').

Alla fine del tutorial avrai già creato la tua prima piccola aggiunta/modifica al gioco originale ('mod'), e sarai pronto a distribuirla a tutti coloro che ancora apprezzano la saga del figlio di Bhaal.

[III.1 I 'Banter'](#)

I 'banter' sono quei dialoghi che compaiono più o meno casualmente durante una partita a Baldur's Gate 2, nei quali gli interlocutori sono due o più 'npc'. Piccola premessa: onde evitare possibili fraintendimenti con altri tipi di dialogo o file, ho preferito non utilizzare sinonimi per la parola 'banter'; essa sarà quindi ripetuta più e più volte in questo tutorial.

In realtà non esistono 'banter' casuali; potrai infatti non sapere in quale ora precisa esso inizierà, ma saprai che al verificarsi di determinate 'condizioni', da te scelte, esso avrà luogo. Tutto ciò è però rimandato al paragrafo successivo; ricorda solo che sono le 'condizioni' a determinare la casualità del 'banter' stesso.

Ora la parte più difficile, il 'banter'. Prima di iniziare a battere sulla tastiera le linee di testo, il mio suggerimento è di pensare al dialogo tra i personaggi; quali interlocutori far intervenire, l'argomento, le eventuali battute ad effetto e la sua conclusione dovrebbero preoccupare più di qualsiasi cosa che puoi apprendere da questo tutorial.

Crea quindi un file testuale di tipo .txt (pulsante destro del mouse sul desktop > nuovo > documento di testo), rinominalo come 'setup.d'. e aprilo con un programma di scrittura (word, wordpad, notepad, etc, per il poco che si fa basta anche l'ultimo dei tre).

[III.2 La variabile](#)

Ogni singolo 'banter' ha bisogno di una 'variabile', una particolare 'condizione', che lo distingue da tutti gli altri 'banter' e che permette al motore del videogioco di sapere quale 'banter' deve essere visualizzato. Ecco l'esempio di una 'variabile':

`Global("BII#MinscNalia","LOCALS",1)`

Il nome della 'variabile', che distingue questo 'banter' da tutti gli altri, è "BII#MinscNalia".

'BII' sta semplicemente per Banter In Italiano, '#' è un separatore e 'MinscNalia' sono i due interlocutori.

In seguito, a 'banter' iniziato, sarà necessario impostare un valore numerico diverso per la 'variabile' onde evitare possibili ripetizioni del 'banter'. Ecco un esempio dell'impostazione di un valore numerico diverso della 'variabile':

`IncrementGlobal("BII#MinscNalia","LOCALS",1)`

III.3 Le linee di testo

Devi sapere che esistono diversi file .dlg per ogni 'npc'; tutti contengono dialoghi, alcuni con il figlio di Bhaal, altri di vario genere, ma quelli nei quali si trovano i 'banter' sono caratterizzati dalla loro iniziale, la 'B'. Riporto qui sotto l'elenco completo dei file .dlg che iniziano per 'B' e lo 'npc' al quale fanno riferimento (la lista sotto 'FILE' sono i .dlg di SoA, gli altri di ToB):

	FILE	25FILE
AERIE	BAERIE	BAERIE25
ANOMEN	BANOMEN	BANOME25
CERND	BCERND	BCERND25
EDWIN	BEDWIN	BEDWIN25
HAERDALIS	BHAERDA	BHAERD25
JAHEIRA	BJAHEIR	BJAHEI25
JAN	BJAN	BJAN25
KELDORN	BKELDOR	BKELDO25
KORGAN	BKORGAN	BKORGA25
MAZZY	BMAZZY	BMAZZY25
MINSC	BMINSC	BMINSC25
NALIA	BNALIA	BNALIA25
VALYGAR	BVALYGA	BVALYG25
VICONIA	BVICONI	BVICON25
YOSHIMO	BYOSHM	BYOSHM
IMOEN2	IMOEN2J	BIMOEN25
SAREVOK	NONE	BSAREV25

Salvo per la prima battuta del 'banter', leggermente diversa dalle altre, le successive sono molto semplici da aggiungere. Ecco un esempio:

`== BMINSC ~Scrivi tra ondine ciò che vuoi Minsc pronunci~`

Continuando nella riga inferiore con un `'= ~Testo~'` se vuoi che sia ancora lo stesso 'npc' a parlare o con un nuovo `'== BNALIA ~Testo~'` se vuoi che cambi il personaggio (nell'esempio Nalia).

Il 'banter' si conclude con un `'EXIT'` sul fondo.

III.4 L'esempio

Se tutto ciò ti è sembrato complicato, ecco un esempio che ti farà sicuramente ricredere:

```

CHAIN IF ~Global("BII#MinscNalia","LOCALS",1)~ THEN BNALIA ruotacricio
~Ho sentito parlare un gran bene delle ruote per criceti; sembra infatti che allungino la vita del roditore se utilizzate
giornalmente.~
DO ~IncrementGlobal("BII#MinscNalia","LOCALS",1)~
== BMINSC ~Davvero? Devo procurarmene una! Gli anni iniziano a farsi sentire anche per Boo.~
== BNALIA ~A dire il vero ne ho già costruita una. L'avevo messa nello zaino, ma ora non la trovo. Lascia che cerchi un
attimo...~
= ~Ed eccola! Tieni Minsc, ops, volevo dire Boo.~
== BMINSC ~Grazie piccola donna. Minsc e Boo ti sono molto grati, soprattutto Boo, vero Boo?~
EXIT

```

'CHAIN IF' deve trovarsi all'inizio di ogni 'banter' ed indica che ciò che si trova immediatamente dopo tra '~~' è 'la variabile'. 'THEN' indica invece l'inizio del dialogo vero e proprio. 'ruotacricio' è solo una indicazione per l'autore del 'banter'. Il 'DO ~~' dopo la prima battuta di Nalia serve per impostare un valore numerico diverso per la 'variabile'.

Vai quindi a capo nel file 'setup.d' e continua con altri 'banter'.

III.5 Il "terzo incomodo"

Nulla vieta che un 'banter' sia fra più di due 'npc'. Ricorda però che se le battute di tutti e tre sono essenziali, la mancanza di uno solo degli 'npc' non farà comparire il 'banter' durante il gioco. Per questo il mio suggerimento è di aggiungere un terzo (o anche quarto, quinto, sesto, etc) "incomodo" che commenta le battute tra i due 'npc' principali senza però fornire un contributo essenziale.

Tutto ciò è molto facile da fare, basta aggiungere tra 'B.dlg' e la battuta tra '~~' un:

```
IF ~IsValidForPartyDialog("nometerzoincomodo")~ THEN
```

L'unica eccezione per i nomi del terzo incomodo è Imoen. Per lei devi aggiungere "Imoen2". Esempio (allungando il precedente 'banter'):

```

...
= ~Ed eccola! Tieni Minsc, ops, volevo dire Boo.~
== BMINSC ~Grazie piccola donna. Minsc e Boo ti sono molto grati, soprattutto Boo, vero Boo?~
== BJAHEIR IF ~IsValidForPartyDialog("Jaheira")~ THEN ~I criceti dovrebbero essere liberi di correre nei prati, non
costretti in quei cosi. Quale sarà il prossimo passo? Una gabbia forse?~
EXIT

```

Se Jaheira non è in gruppo, il 'banter' si concluderà con l'ultima battuta di Minsc; in caso contrario, l'ultima parola spetterà a Jaheira.

Volendo si può anche far rispondere uno dei due 'npc' principali all'eventuale commento del terzo incomodo (Jaheira); basterà continuare ad inserire 'IF ~IsValidForPartyDialog("Jaheira")~ THEN'. Esempio:

```

...
= ~Ed eccola! Tieni Minsc, ops, volevo dire Boo.~
== BMINSC ~Grazie piccola donna. Minsc e Boo ti sono molto grati, soprattutto Boo, vero Boo?~
== BJAHEIR IF ~IsValidForPartyDialog("Jaheira")~ THEN ~I criceti dovrebbero essere liberi di correre nei prati, non
costretti in quei cosi. Quale sarà il prossimo passo? Una gabbia forse?~
== BMINSC IF ~IsValidForPartyDialog("Jaheira")~ THEN ~Tranquilla, amica mia, nessuno potrà mai contenere lo
spirito indomito di Boo; sia una perfido mago che ruba l'anima agli amici di Minsc e Boo o una esile gabbia.~
EXIT

```

Volendo ancora, si può far pronunciare una nuova frase ad uno dei due 'npc' principali solo se il terzo incomodo non è presente. Per far ciò basterà aggiungere un punto esclamativo prima di 'IsValidForPartyDialog' (un punto esclamativo davanti a qualsiasi 'condizione' la pone al contrario). Esempio:

```

...
= ~Ed eccola! Tieni Minsc, ops, volevo dire Boo.~

```

```

== BMINSC ~Grazie piccola donna. Minsc e Boo ti sono molto grati, soprattutto Boo, vero Boo?~
== BJAHEIR IF ~IsValidForPartyDialog("Jaheira")~ THEN ~I criceti dovrebbero essere liberi di correre nei prati, non costretti in quei cossi. Quale sarà il prossimo passo? Una gabbia forse?~
== BNALIA IF ~!IsValidForPartyDialog("Jaheira")~ THEN ~Per fortuna Jaheira non è qui. Immagini quale malvagità avrebbe visto celata dietro questa piccola ruota in legno?~
== BMINSC IF ~IsValidForPartyDialog("Jaheira")~ THEN ~Tranquilla, amica mia, nessuno potrà mai contenere lo spirito indomito di Boo; sia una perfido mago che ruba l'anima agli amici di Minsc e Boo o una esile gabbia.~
EXIT

```

III.6 Altri casi di battute particolari

Le battute non necessarie possono anche essere degli stessi 'npc' principali. In questo caso la 'condizione' non sarà più 'IsValidForPartyDialog("nometerziincomodo)", ma una delle altre decine possibili. Esempio (morte di Bodhi):

```

...
= ~Ed eccola! Tieni Minsc, ops, volevo dire Boo.~
== BMINSC IF ~Grazie piccola donna. Minsc e Boo ti sono molto grati, soprattutto Boo, vero Boo?~
== BMINSC IF ~!Dead("Bodhi")~ THEN ~Boo, che ne dici di far un pò di esercizio? Ho sentito che i vampiri sono molto veloci; dovrai essere pronto a scattare se ci sarà bisogno di te.~
...

```

Quest'ultima frase comparirà solo se Bodhi non è ancora morta.

Ecco una lista delle 'condizioni' più comuni:

Class(Player1,FIGHTER)	Vuol dire
HPPercentGT(Player1,75)	Il figlio di Bhaal è un guerriero (CLERIC per chierico, THIEF per ladro, etc)
HPPercentGT("Jaheira",75)	Il figlio di Bhaal ha più del 75% di punti ferita
ReputationGT(Player1,15)	Jaheira ha più del 75% di punti ferita (o Minsc, Nalia, etc)
TimeOfDay(Day)	La reputazione del gruppo è superiore a 15
Gender("Edwin",FEMALE)	E' giorno (o Night per notte)
PartyGoldGT(10000)	Edwin è una donna
NumInPartyGT(3)	Il gruppo ha più di 10000 m.o.
AreaType(FOREST)	Il gruppo è formato da più di tre elementi
LevelGT("Minsc",15)	Ti trovi in un bosco (o City per città, o DUNGEON, etc)
Global("Chapter",GLOBAL",2)	Minsc è un ranger di livello superiore al 15
GlobalGT("Chapter",GLOBAL",2)	Ti trovi nel capitolo 2 del gioco
AreaCheck("codice area")	Ti trovi dal capitolo 3 in avanti del gioco
PartyHasItem("codice oggetto")	Il gruppo si trova nell'area indicata. <u>Qui</u> puoi trovarne un elenco delle aree
	Il gruppo ha l'oggetto indicato. <u>Qui</u> puoi trovarne un elenco degli oggetti

Player1 e il nome dello 'npc' sono sempre intercambiabili.

Altre importanti 'condizioni' sono:

Global (intesa come qualsiasi 'variabile' del gioco)
HaveSpell

Per queste ultime però il discorso si fa più complicato. Per la necessità di ulteriori e lunghe spiegazioni esse non verranno trattate in questo tutorial. Tieni però a mente che già con le solo 'condizioni' comuni potrai sbizzarrirti nello scrivere i 'banter' nelle più disparate situazioni.

IV. Gli script

Abbandona ora il file 'setup.d' e vai a creare tanti file testuali 'nomenpc.baf' quanti sono gli 'npc' che hanno una battuta d'esordio nei 'banter' che hai scritto in 'setup.d.' (le uniche due eccezioni sono Haer-Dalis, per il quale dovrai scrivere

'haerdali.baf' ed Imoen, per la quale dovrai scrivere 'imoen2.baf').

In esso dovrai scrivere le 'condizioni' per le quali il 'banter' avrà luogo durante il gioco. A differenza delle 'condizioni' di cui poco sopra, queste si differenziano perchè essenziali per il 'banter'. Quelle del sotto paragrafo 'Altri casi di battute particolari', sono invece solo qualcosa in più e, come già ripetuto, possono anche non aver luogo facendo semplicemente terminare prima il 'banter'.

Nei file .baf, per ogni banter, ci dovrà essere una struttura del tipo

```
IF
..condizioni..
THEN
    RESPONSE #100
..fai partire il 'banter'..
END
```

Tradotto: se le 'condizioni' si verificano, allora fai partire il 'banter'.

Tra le condizioni ci dovranno sempre essere:

```
InParty(Myself)
InParty("nomealtronpc")
See("nomealtronpc")
!StateCheck("nomealtronpc",STATE_SLEEPING)
!Dead("nomealtronpc")
CombatCounter(0)
```

Tradotto: se i due 'npc' del 'banter' sono nel gruppo, se sono nel loro campo visivo, se non stanno dormendo, se non sono morti e se non ci sono battaglie in corso.

Più le 'condizioni' che identificano il 'banter', le quali sono la 'variabile' di identificazione del banter (nell'esempio: "BII#MinscNalia") e le altre riportate alla fine del paragrafo precedente. In totale, avrai bisogno di almeno due 'condizioni' identificative oltre a quelle obbligatorie; in caso contrario, il 'banter' partirà non appena i due 'npc' faranno parte del gruppo. Esempio (Nalia.baf):

```
IF
    InParty(Myself)
    InParty("Minsc")
    See("Minsc")
    !StateCheck("Minsc",STATE_SLEEPING)
    !Dead("Minsc")
    CombatCounter(0)
    Global("BII#MinscNalia","LOCALS",0)
    AreaType(FOREST)
    LevelGT("Minsc",12)
THEN
    RESPONSE #100
        IncrementGlobal("BII#MinscNalia","LOCALS",1)
        Interact("Minsc")
END
```

Grazie alle due 'condizioni' finali hai reso il 'banter' casuale; non saprai in quale momento esatto esso partirà (dopo un ora di gioco? Due? Venti?), ma esso avverrà sicuramente quando Minsc sarà almeno di livello 12 e il gruppo si troverà in una foresta.

Un ultimo e importante tipo di 'condizione', questa volta utilizzabile solamente negli script, è il tempo. Volendo puoi infatti porre anche come unica 'condizione' il trascorrere di un determinato numero di secondi (veri) o di giorni (nel gioco) per il verificarsi del 'banter'. Per far ciò avrai bisogno di due strutture **IF - THEN - END**; la prima servirà a far partire il timer, la seconda il 'banter'. Esempio:

```
IF
    InParty(Myself)
```

```

        InParty("Minsc")
        Global("BII#MinscNalia", "LOCALS", 0)
THEN
        RESPONSE #100
            SetGlobal("BII#MinscNalia", "LOCALS", 50)
            RealSetGlobalTimer("BII#MinscNaliaT", "LOCALS", 7200)
END

```

```

IF
    InParty(Myself)
    InParty("Minsc")
    See("Minsc")
    !StateCheck("Minsc", STATE_SLEEPING)
    !Dead("Minsc")
    CombatCounter(0)
    Global("BII#MinscNalia", "LOCALS", 50)
    RealGlobalTimerExpired("BII#MinscNaliaT", "LOCALS")
THEN
    RESPONSE #100
        SetGlobal("BII#MinscNalia", "LOCALS", 1)
        Interact("Minsc")
END

```

Tradotto (primo IF - THEN - END): se Minsc e Nalia si trovano nello stesso gruppo, allora fai partire il timer "BII#MinscNaliaT" di 7200 secondi (due ore).

Tradotto (secondo IF - THEN - END): se Minsc e Nalia si trovano nello stesso gruppo, se sono nel loro campo visivo, se non stanno dormendo, se non sono morti, se non ci sono battaglie in corso e se sono passati 7200 secondi, allora fai partire il 'banter'.

L'altro tipo di timer ha come unità di misura i giorni così come segnalati nel calendario del gioco durante la partita. Invece di 'RealSetGlobalTimer("BII#MinscNaliaT", "LOCALS", 7200)' avrai:

```
'SetGlobalTimer("BII#MinscNaliaT", "LOCALS", ONE_DAY)'
```

per un timer di un giorno, 'TWO_DAYS' per uno di due giorni e così via. Invece di 'RealGlobalTimerExpired("BII#MinscNaliaT", "LOCALS")' avrai:

```
'GlobalTimerExpired("BII#MinscNaliaT", "LOCALS")'
```

Nota infine che il timer non deve necessariamente partire quando i due 'npc' si incontrano nel gruppo; al primo IF - THEN - END puoi aggiungere una 'condizione' che posticipa il conto alla rovescia. Esempio (per far partire il timer solo dal capitolo 3):

```

IF
    InParty(Myself)
    InParty("Minsc")
    GlobalGT("Chapter", "GLOBAL", 2)
    Global("BII#MinscNalia", "LOCALS", 0)
THEN
    RESPONSE #100
        SetGlobal("BII#MinscNalia", "LOCALS", 50)
        RealSetGlobalTimer("BII#MinscNaliaT", "LOCALS", 7200)
END

```

Una volta scritti tutti i 'banter' in 'setup.d' e terminati tutti gli script nei .baf, potrai passare alla creazione del setup.

V. Creare l'installazione

Stai ora per creare il setup che permetterà a chiunque altro di installare il tuo mod nella sua partita a Baldur's Gate 2.

Per prima cosa devi sapere che quasi tutti i mod per Baldur's Gate 2 sono scritti in un linguaggio detto weidu. Tutto ciò che hai appreso da questo tutorial sono nozioni di weidu.

Scarica ora l'intero programma weidu da [qui](#). Apri lo .zip con winrar o winzip ed estrai il file weidu.exe. Rinominalo quindi come 'Setup-NomedelMod.exe' (per l'esempio sarà Setup-ItalianBanter.exe) e mettilo da parte.

Crea ora una cartella e chiamala 'IBanter'. Al suo interno posiziona 'setup.d' e tutti i '.baf'. Crea quindi una nuova cartella (sempre all'interno di 'IBanter') e chiamala 'Backup'.

Crea ancora un nuovo file testuale e rinominalo come 'Setup-ItalianBanter.tp2'. Apri il file e scrivi (dopo '/' trovi la descrizione dell'ordine):

```
BACKUP ~IBanter\Backup~
```

```
AUTHOR ~Scrivi qui il tuo nome/nick e possibilmente anche la tua e-mail~
```

```
BEGIN ~Vuoi installare il mod Italian Banter?~ // questo è ciò che si leggerà durante l'installazione
```

```
COMPILE ~IBanter/setup.d~ // questo è l'ordine con cui si aggiungono le battute dei 'banter'
```



```
EXTEND_TOP ~aerie.bcs~ ~IBanter/aerie.baf~ // questo è l'ordine con cui si integrano gli script degli 'npc'
```

```
EXTEND_TOP ~altronpc.bcs~ ~IBanter/altronpc.baf~
```

```
....
```

Salva il file e chiudi.

Ora avrai sullo stesso livello la cartella 'IBanter', e i due file 'Setup-ItalianBanter.tp2' e 'Setup-ItalianBanter.exe'.



Comprimi i tre con winrar, trova un posto che ospiti il tuo mod (se vuoi contattami su rtt-mod@libero.it) e annuncia al mondo la tua opera sul forum di www.spellholdstudios.com

Complimenti, il tuo mod è appena finito!

VI. La traduzione

Il tuo mod non è "esattamente" finito. Sebbene il mod sia ora funzionante, esso è disponibile solo nella nostra amata

lingua. Weidu permette però di avere installazioni multilingua. Per far ciò devi copiare 'setup.d' all'interno della cartella 'weidu' (il collegamento per scaricarla è fornito poco più su), creare un nuovo file testuale e rinominarlo come 'trad.bat'.

Apri il file con notepad e scrivi:

```
weidu --traify setup.d
```

Salva, chiudi il file e lancialo (nel raro caso in cui si verificasse un errore, sposta tutti i file contenuti nella cartella 'weidu' dove Baldur's Gate 2 è installato, quindi rilancia 'trad.bat'). Nella cartella comparirà il nuovo file 'setup.tra'. Al suo interno troverai le battute che prima erano contenute nel solo 'setup.d', e nel nuovo 'setup.d' troverai le battute sostituite da '@NUMERI'. Utilizzando '--traify' per il 'banter' preso come esempio in questo tutorial, il nuovo 'setup.d' apparirà così:

```
CHAIN IF ~Global("BII#MinscNalia","LOCALS",1)~ THEN BNALIA ruotacricio
@0
DO ~IncrementGlobal("BII#MinscNalia","LOCALS",1)~
== BMINSC @1
== BNALIA @2
= @3
== BMINSC @4
EXIT
```

E il 'setup.tra' così:

```
@0 = ~Ho sentito parlare un gran bene delle ruote per criceti; sembra infatti che allungino la vita del roditore se
utilizzate giornalmente.~
@1 = ~Davvero? Devo procurarmene una! Gli anni iniziano a farsi sentire anche per Boo.~
@2 = ~A dire il vero ne ho già costruita una. L'avevo messa nello zaino, ma ora non la trovo. Lascia che cerchi un
attimo...~
@3 = ~Ed eccola! Tieni Minsc, ops, volevo dire Boo.~
@4 = ~Grazie piccola donna. Minsc e Boo ti sono molto grati, soprattutto Boo, vero Boo?~
```

Aggiungi in fondo a 'setup.tra' anche:

```
@1000 = ~Vuoi installare il mod Italian Banter?~
```

Sostituisci adesso il vecchio 'setup.d' all'interno della cartella 'IBanter' con quello nuovo e metti 'setup.tra' in una nuova cartella chiamata 'italian' sempre all'interno della cartella 'IBanter'.

Apri 'Setup-ItalianBanter.tp2' e modificalo come indicato qui sotto

```
BACKUP ~IBanter\Backup~
AUTHOR ~Scrivi qui il tuo nome/nick e possibilmente anche la tua e-mail~
```

```
LANGUAGE ~Italiano~
~italian~
~IBanter/italian/setup.tra~
```


```
BEGIN @1000 // questo è ciò che si leggerà durante l'installazione
```

```
COMPILE ~IBanter/setup.d~ // questo è l'ordine con cui si aggiungono le battute dei 'banter'
```

```
EXTEND_TOP ~aerie.bcs~ ~IBanter/aerie.baf~ // questo è l'ordine con cui si integrano gli script degli 'npc'
EXTEND_TOP ~altronpc.bcs~ ~IBanter/altronpc.baf~
....
```

Se in futuro vorrai tradurre il tuo mod, ad esempio, in inglese, non dovrai far altro che creare la cartella 'english' all'interno della cartella 'IBanter' e lì posizionare il 'setup.tra' tradotto. Dovrai poi aggiungere nel ' Setup-ItalianBanter.tp2', tra '~IBanter/italian/setup.tra~' e 'BEGIN @1000', quanto segue:

```
LANGUAGE ~English~
~english~
~IBanter/english/setup.tra~
```

VII. Conclusione

Ora il tutorial può dirsi finalmente concluso. Per qualsiasi dubbio puoi contattarmi su rtt-mod@libero.it Il mio sito internet è www.rtt.altervista.org